content: -- LinkTool for InDesign
  -- version 1.1
  
  -- created by medul6, Michael Heck, 2014
  -- open sourced on August 12th, 2014 on Github > check the LICENSE.txt and README.md in the repository for detailed information
  -- https://github.com/medul6/indesign-link-tool
  
  -- •••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••
  
  property functionChoice : {"Linkcheck AUS"}
  property fileName : "Beispiel.tif"
  
  -- •••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••
  
  my whatFunction()
  
  -- •••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••
  
  on whatFunction()
  	set functionChoice to choose from list {"Linkcheck AUS", "Links aktualisieren", "Linkcheck EIN", "Gehe zu Verknüpfung (über Name)"} default items functionChoice with prompt "Funktion wählen:" OK button name "Do it!"
  	
  	if the functionChoice = {"Linkcheck AUS"} then
  		my linkCheckOff()
  	else if the functionChoice = {"Links aktualisieren"} then
  		my updateLinks()
  	else if the functionChoice = {"Linkcheck EIN"} then
  		my linkCheckOn()
  	else if the functionChoice = {"Gehe zu Verknüpfung (über Name)"} then
  		my goToNamedLink()
  		
  	end if
  end whatFunction
  
  -- •••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••
  
  on linkCheckOff()
  	tell application id "com.adobe.InDesign"
  		set check links at open of linking preferences to false
  		set find missing links at open of linking preferences to false
  	end tell
  	my theEnd()
  end linkCheckOff
  
  -- •••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••
  
  on updateLinks()
  	tell application id "com.adobe.InDesign"
  		set openDocuments to every document
  		
  		repeat with x from 1 to count openDocuments -- this iterates through all open documents
  			tell item x of openDocuments
  				set missingLinks to (every link whose status is link out of date)
  				repeat with aMissingLink in missingLinks
  					update aMissingLink
  				end repeat
  			end tell
  		end repeat
  		
  	end tell
  	my theEnd()
  end updateLinks
  
  -- •••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••
  
  on linkCheckOn()
  	tell application id "com.adobe.InDesign"
  		set check links at open of linking preferences to true
  		set find missing links at open of linking preferences to true
  	end tell
  	my theEnd()
  end linkCheckOn
  
  -- •••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••
  
  on goToNamedLink()
  	tell application id "com.adobe.InDesign"
  		
  		display dialog "Vollständiger Name des Links bitte angeben." & return & "(Wählt nur die erste Instanz in der aus," & return & "falls die Datei mehrfach platziert ist)" default answer fileName buttons {"Abbrechen", "Weiter!"} default button "Weiter!"
  		if button returned of result is "Weiter!" then
  			set fileName to (text returned of result)
  			set tempFilteredLinkList to (every link of active document whose name is fileName)
  			if (count tempFilteredLinkList) > 0 then
  				set selection of active document to parent of parent of first item of tempFilteredLinkList
  				tell active window to zoom given fit spread
  				my theEnd()
  			else
  				display alert "Nicht gefunden ..." as informational
  				--informational: the standard alert dialog
  				--warning: the alert dialog dialog is badged with a warning icon
  				--critical: currently the same as the standard alert dialog
  			end if
  		else
  			return
  		end if
  	end tell
  end goToNamedLink
  
  -- •••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••
  
  on theEnd()
  	--display dialog "Fertig!" buttons "OK" default button "OK" giving up after 1
  	--say "OK!" using "Zarvox" --"Zarvox"
  	say some item of {"OK!", "Yo!", "Done!", "Fine!", "Roger!", "Yes!"} using "Zarvox" volume 0.5 speaking rate 300 --pitch 10 -- pitch from 0 to 127
  end theEnd
  
  -- •••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••••
  